EGRE 426

Lab 5

Design of FSM Control Unit for the Non-pipelined KIPS

The purpose of this lab is to understand how to implement a finite state machine (FSM) control unit for the non-pipelined version of the KIPS processor discussed in class. 

A complete lab report will be due at the beginning of the next lab next Monday. Each person must work on this lab individually. Provided each person makes significant contributions to completing the lab, you may work on the lab with your lab partner and turn in a single joint lab report. In your lab report discuss the additions and changes you made and the reasons for your design decisions. Do not change the entity for the cu.vhd file without approval from the instructor. You must demonstrate your design during Monday’s lab period, and you must turn in your lab report at the time of the demonstration. 

Three files (pkg_npkips.vhd, npkips_cu.vhd, and tb_cu.vhd) are provided.

1. pkg_npkips.vhd – contains general definitions.

2. tb_cu.vhd – is the test bench

3. npkips_cu.vhd – implements only the state portion of FSM of the control unit.  You do not need to be concerned with the register transfer output portion of the FSM.

The simulation waveform is shown below. Note that signal “inst” is used to show the mnemonic for the current instruction.

Make the necessary additions to the above programs to include the “addi” and “jr” instructions. Your revised test bench must include these new instructions. Only model the state transitions. You are not expected to include the register transfers associated with each state; however, add your new instructions, along with the associated register transfer operations, to both the state diagram and HDL description below.

Note that R_Type instructions include “add”, “sub”, “and”, and “or” instructions all of which have an opcode of 0x00 and use the ALU with different functions. R_type does not include the jr instruction which also has an opcode of 0x00. The func bits are used to make the distinction. Refer to appendix A on the books CD. 

IF Reset then Goto S0
S0: 
PC ( PC + 4, IR ( M(PC), goto S1

S1:
A ( Rs, B ( Rt,

ALUout ( PC + (n||00
i.e. ALUout ( ALU := (SA := PC) + (SB := 4)


if op = J goto S9, if op = BEQ goto S8, if op = R-type goto S6


if op = LW or SW goto S2

S2:
ALUout ( A + (ofst


i.e.  ALUout ( ALU := (SA := A) + (SB := (ofst)


if op = LW goto S3


if op = SW goto S5

S3:
MDR ( M(ALUout), goto S4

S4:
Rt (MDR, goto S0

S5:
M(ALUout) ( B, goto S0

S6:
ALUout ( A * B
i.e. ALUout ( ALU := (SA := A) * (SB := B)

 goto S7

S7:
Rd ( ALUout, goto S0

S8:
if A = B then PC ( ALUout


Goto S0

S9:
PC ( PC(31..28)||addr||00 

goto S0

-- PKG_NPKIPS.VHD Version 2.0 

LIBRARY IEEE;

USE work.all;

USE IEEE.Std_Logic_1164.all; 

use IEEE.STD_LOGIC_ARITH.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL; 

PACKAGE pkg_npkips IS

    TYPE A_State IS (S0, S1, S2, S3, S4, S5, S6,

                     S7, S8, S9, S10, S11, S12);

    SUBTYPE Opcode is STD_LOGIC_vector(5 downto 0);

    SUBTYPE Func   is STD_LOGIC_vector(5 downto 0);

-- Mapping of instructions to opcodes. See Apendix A.

    CONSTANT Op_lw      :Opcode := B"100011";   -- 0x23          

    CONSTANT Op_sw      :Opcode := B"101011";   -- 0x2b           

    CONSTANT Op_beq     :Opcode := B"000100";   -- 0x04            

    CONSTANT Op_addi    :Opcode := B"001000";   -- 0x08

    CONSTANT Op_andi    :Opcode := B"001100";   -- 0x0c

    CONSTANT Op_ori     :Opcode := B"001101";   -- 0x0d

    CONSTANT Op_xori    :Opcode := B"001110";   -- 0x0e

    CONSTANT Op_slti    :Opcode := B"001010";   -- 0x0a

    CONSTANT Op_or      :Opcode := B"000000";   -- 0x00 see func            

    CONSTANT Op_slt     :Opcode := B"000000";   -- 0x00 see func            

    CONSTANT Op_j       :Opcode := B"000010";   -- 0x02 

    CONSTANT Op_jr      :Opcode := B"000000";   -- 0x00 see func

    CONSTANT Op_R_Type  :Opcode := B"000000";   -- 0x00 use func

-- The instructions below have an op code of "000000"

    CONSTANT Func_add   :Func   := B"100000";   -- 0x20    

    CONSTANT Func_sub   :Func   := B"110100";   -- 0x34    

    CONSTANT Func_and   :Func   := B"000100";   -- 0x36

    CONSTANT Func_or    :Func   := B"010101";   -- 0x25    

    CONSTANT Func_slt   :Func   := B"011010";   -- 0x42

    CONSTANT Func_jr    :Func   := B"001000";   -- 0x08

    CONSTANT Func_xor   :Func   := B"100110";   -- 0x38

    CONSTANT Func_nor   :Func   := B"100111";   -- 0x39

END PACKAGE pkg_npkips;

NPKIPS.VHD

LiBRARY IEEE;

USE work.ALL;

USE IEEE.std_logic_1164.ALL;

use IEEE.STD_LOGIC_ARITH.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL;

use pkg_npkips.all;

ENTITY CU IS

    PORT (CLK   : IN  STD_LOGIC;

          RES   : IN  STD_LOGIC;

          OP    : in  STD_LOGIC_VECTOR(5 downto 0);

          FUNC  : in  STD_LOGIC_VECTOR(5 downto 0);

          S     : OUT A_STATE

         );

END ENTITY CU;

ARCHITECTURE arch OF CU IS

    SIGNAL STATE, Next_State : A_State;

BEGIN

    S <= STATE;

    PROCESS(State, op)

BEGIN

    CASE State IS

        WHEN S0 =>

            Next_State <= S1;

        WHEN S1 =>

            CASE OP IS

                WHEN op_lw | op_sw =>

                    Next_State <= S2;

                WHEN op_R_type =>

                    Next_State <= S6;

                WHEN op_BEQ =>

                    Next_State <= S8;

                WHEN op_J =>

                    Next_State <= S9;

                WHEN others =>

                    NEXT_STATE <= S0;

            END CASE;

        WHEN S2 =>

            IF OP = op_lw then

                Next_State <= S3;

            ELSE

                Next_State <= S5;

            END IF;

        WHEN S3 =>

            NEXT_STATE <= S4;

        WHEN S4 | S5 | S7 | S8 | S9 =>

            NEXT_STATE <= S0;

        WHEN S6 =>

            NEXT_STATE <= S7;

        WHEN others =>

            Next_State <= S0;

        END CASE;

    END PROCESS;

    -- 

    PROCESS(CLK, RES)

    BEGIN

        IF (RES = '1') THEN

            State <= S0;

        ELSIF clk'event and clk = '1' THEN

            State <= Next_State;

        END IF;

    END PROCESS;

END arch;

TB_CU.VHD

-----------------------------------------------------------------------

-- tb_cu.vhd - Test bench for npkips cu.

-----------------------------------------------------------------------

LiBRARY IEEE;

USE work.ALL;

USE IEEE.std_logic_1164.ALL;

use IEEE.STD_LOGIC_ARITH.ALL;

use IEEE.STD_LOGIC_UNSIGNED.ALL;

use pkg_npkips.all;

ENTITY TB IS

END ENTITY TB;

ARCHITECTURE TEST of TB is

    signal Reset:   std_logic := '1';

    signal CLK:     std_logic := '0';

    signal S: A_STATE; -- Type A_STADE is defined in PKG_NPKIPS.VHD

    signal op: std_logic_vector(5 downto 0);

    signal func: std_logic_vector(5 downto 0) := "100000";

    Type Inst_type is (J, BEQ, R_Type, LW, SW);

    signal inst: inst_type;

BEGIN

    U1: ENTITY work.cu(arch)

        PORT MAP

        (

         CLK => CLK,

         RES => Reset,

         OP => OP,

         FUNC => FUNC,

         S => S

        );

    CLK <= not CLK after 5 ns when Reset = '0' else '0';

    Reset <= '0' after 10 ns;

    process 

    begin

        op <= op_lw;

        wait on S;

        wait until s = s0;

        op <= op_sw;

        wait on S;

        wait until s = s0;

        op <= op_j;

        wait on S;

        wait until s = s0;

        op <= op_R_type;

        wait on S;

        wait until s = s0;

        op <= op_beq;

        wait on S;

        wait until s = s0;

    end process;

-- Set inst to mnemonic for current instruction

    process(op)

    begin

        case op is

            when op_lw => inst <= LW;

            when op_sw => inst <= sw;

            when op_R_type => inst <= R_type;

            when op_J => inst <= J;

            when op_beq => inst <= BEQ;

            when others => null;

        end case;

    end process;    

END ARCHITECTURE TEST;

